



Συστήματα Μικροϋπολογιστών

Τρίτη ομάδα ασκήσεων

Σπουδαστές

Κατσάμπουλα Χριστίνα Σοφία (Α.Μ.: 03114910)

Παπασκαρλάτος Αλέξανδρος (Α.Μ.: 03111097)

Ημερομηνία Υποβολής Αναφοράς: 1 Μαΐου 2018

Άσκηση 1i

IN 10H

START:

LDA 2000H

MVI B,08H

LOOP1: ;Diabazw tous diakoptes apo aristera pros ta deksia

RLC ;mexri na brw to prwto OFF

JNC LEDS ;h mexri na diabasw kai tous 8 (an einai oloi ON)

DCR B

JNZ LOOP1

LEDS: ;An (ksekinwntas apo aristera) o x diakopths einai OFF

MVI A,FFH ;8elw na anapsw ta 9-x LSB LEDS

INR B ;Ayksanoume kata ena to B, giati parakatw

;to prwto run tou loop einai "axrhsto"

LOOP2: ;8eloume "axrhsto" to prwto perasma

RLC ;gia thn periptwsh pou einai ola ON

INR A ;Se ka8e perasma kanoume peristrofh kai

DCR B ;pros8etoume 01H, dhladh 8etoume to LSB sto 1 (arxika htan 0)

JNZ LOOP2

CMA ;Antistrefoume giati ta LEDS einai arnhtikhs logikhs

STA 3000H

JMP START

END

Άσκηση 1ii

IN 10H

START:

MVI A,09H ;Arxikopoioume sto 09H>08H

CALL KIND ;Diabazoume plhktrologio

CPI 09H ;An path8hke kati me kwdiko>08h,

JNC START ;pame pali sthn arxh

CPI 08H ;An path8hke to 8

JZ KEY_8 ;anabosbhnoume ta LEDS

MVI B,07H ;Diaforetika, to sygkrinoume diadoxika

;me olous tous ari8mous apo 7 ews 0

LOOP1:

CMP B

JZ LEDS

DCR B

JNZ LOOP1

LEDS: ;Arxikopoioume tis times

MVI A,80H ;wste to prwto loop na einai "axrhsto"

INR B ;mias kai 8a ginei sigoura

;o elegxos tou counter ginetai sto telos)

LOOP2: ;Kanoume peristrofes analoga me ton ari8mo pou brhkame

RLC

DCR B

JNZ LOOP2

CMA ;Anaboume ta arnhtikhs logikhs LEDS

STA 3000H

JMP START

KEY_8: ;Path8hke to 8

MVI A,00H

LXI B,0200H ;8etoume to B-C sta 0200H=512D

MVI E,14H ;8etoume ton counter sta 14H=20D

LOOP3: ;Anabei kai sbhnei apo 10 fores ta LEDS

STA 3000H

CMA

CALL DELB ;Ka8ysterhsh ish me 512ms

DCR E ;To loop ginetai 20 fores (10 anabei, 10sbhnei)

JNZ LOOP3

JMP START

END

Άσκηση 1iii

```
IN 10H
LXI H,0902H ;8etoume sth 8eseis mnhmhs
MVI M,10H ;0902H ews 0905H
INX H ;ton ari8mo 10H
MVI M,10H ;pou einai o kwdikos tou " "
INX H ;wste telika, otan emfanisoume sthn o8onh
MVI M,10H ;na mhn anapsoun ta 4 prwta pshfia
INX H
MVI M,10H
```

START:

```
MVI D,FEH ;Ssekinaw apo thn prwth grammh
MOV A,D ;FE H = 11111110 b
```

```
;Grammh 0: INSTR STEP, FETCH PC
STA 2800H
LDA 1800H ;Diabazw tis sthles
ANI 07H ;mhdenizw ta 5 MSB
MVI C,86H ;INSTR STEP: kwdikos 86H
CPI 06H ;Sygkrinw me to (00000)110 (1o plhktro)
JZ PRINT ;An path8hke, kanoume print
MVI C,85H ;FETCH PC:kwdikos 85H
CPI 05H ;Sygkrinw me to (00000)101 (2o plhktro)
JZ PRINT
```

```
MOV A,D ;Ka8e fora mono ena bit einai 0
RLC ;Me peristrofh aristera,
MOV D,A ;epilegoume thn epomenh grammh
```

```
;Grammh 1: RUN, FETCH REG, FETCH ADRS
STA 2800H
LDA 1800H
ANI 07H
MVI C,84H ;RUN: kwdikos 84H
CPI 06H ;Sygkrinw me to 110 (1o plhktro)
JZ PRINT
MVI C,80H ;FETCH REG: kwdikos 80H
CPI 05H ;Sygkrinw me to 101 (2o plhktro)
JZ PRINT
MVI C,82H ;FETCH ADRS: kwdikos 82H
CPI 03H ;Sygkrinw me to 011 (3o plhktro)
JZ PRINT
```

```
MOV A,D
RLC
MOV D,A
```

;Grammh 2: 0, STORE/INCR, DECR

```

STA 2800H
LDA 1800H
ANI 07H
MVI C,00H    ;0: kwdikos 00H
CPI 06H
JZ PRINT
MVI C,83H    ;STORE/INCR: kwdikos 83H
CPI 05H
JZ PRINT
MVI C,81H    ;DECR: kwdikos 81H
CPI 03H
JZ PRINT

```

```

MOV A,D
RLC
MOV D,A

```

```

;Ta pragmata ginontai pio eykola
;O kwdikos tou ka8e plhktrou ayksanetai
;diadoxika apo to 01 ews to 0F
;Opote ylopoioume tis grammes 3-7 me loop

```

```

;Grammes 3 ews 7: plhktra 1 ews F
MVI B,05H    ;Loop counter (5 grammes->5 loops)
MVI C,00H    ;8a exei ton kwdiko tw n antisoixwn plhktrwn

```

LOOP1:

```

STA 2800H
LDA 1800H
ANI 07H
INR C
CPI 06H
JZ PRINT
INR C
CPI 05H
JZ PRINT
INR C
CPI 03H
JZ PRINT

```

```

MOV A,D
RLC
MOV D,A

```

```

DCR B
JNZ LOOP1

```

```

JMP START    ;An den path8hke tipota, pame sthn arxh

```

```

PRINT:        ;Emfanizoume ton antistoixo kwdiko sthn o8onh

```

```
LXI H,0900H
MOV A,C
ANI 0FH      ;Apomonwnw ta 4 LSB
MOV M,A      ;kai ta apo8hkeyw sth 8esh mnhmhs 0900H
INX H
MOV A,C
ANI F0H      ;Apomonwnw ta 4 MSB
RRC          ;Kanw 4 peristrofes
RRC          ;gia na ta balw sta 4 LSB bits
RRC
RRC
MOV M,A      ;Kai ta apo8hkeyw sth 8esh mnhmhs 0901H

LXI D,0900H  ;Arxikopoiw sth 8esh 0900H
CALL STDM    ;kai emfanizw sthn o8onh
CALL DCD

JMP START

END
```

Άσκηση 2

Σχεδιάζουμε σύστημα για μεταφορά δεδομένων από ένα μΥ-Σ 8085 σε ένα άλλο μΥ-Σ 8085. Το μΥ-Σ 1 στέλνει 256 δεδομένα και το μΥ-Σ 2 τα λαμβάνει, τα αποθηκεύει και αποθηκεύει το μικρότερο από αυτά στον καταχωρητή C.

Υποθέτουμε πως και για τα δύο μΥ-Σ, η θύρα 10H χρησιμοποιείται για είσοδο δεδομένων, η θύρα 40H για έξοδο δεδομένων, η θύρα 20H για το SOD και η θύρα 30H για το SID.

Θεωρώ πως η αρχική θέση μνήμης για τον κάθε μΥ-Σ είναι 0900H

:μΥ-S 1 (Χρησιμοποιούμε αυτόν τον μΕ για να στείλουμε δεδομένα)

MVI C,00H ;O C 8a periexei to loop counter gia 256 dedomena
LXI H,0900H ;Estw 0900H h prwth 8esh mnhmhs

START:
MVI A,01H ;O mE 8etei 1 to SOD1 ("etoimos na steilw")
OUT 20H

POLL: ;Elegxw pote to SID1 ginei 1
IN 30H
RRC
JNC POLL

MOV A,M ;8etw to dedomeno mou sth 8yra eksodou 40H
OUT 40H

MVI A,00H ;Kai mhdenizw to SOD1 ("pare ena byte")
OUT 20H

INX H ;Paw sthn epomenh 8esh mnhmhs
DCR C ;Kai meiwnw ton loop counter
JZ TELOS ;An C=0, exoun stalei 256 dedomena, opote teleiwsame

STOP: ;Elegxw pote to SID1 ginei 0
IN 30H
RRC
JC STOP
JMP START

TELOS:
END

:mY-S 2 (Χρησιμοποιούμε αυτόν τον µΕ για να λάβουμε δεδομένα)

LXI H,0900H	;Estw 0900H h prwth 8esh mnhmhs
MVI C,FFH	;O C 8a periexei to mikrotero mexri twra dedomeno
MVI A,00H	;Arxikopoiw to SOD2 sto 0
OUT 20H	
START:	
POLL:	;Elegxw pote to SID2 ginei 1
IN 30H	
RRC	
JNC POLL	
MVI A,01H	;O mE 8etei 1 sto SOD2 ("etoimos na labw")
OUT 20H	
POLL2:	;Elegxw pote to SID2 ginei 0
IN 30H	
RRC	
JC POLL2	
IN 10H	;Lambanw to dedomeno apo th 8yra eisodou 10H
MOV M,A	;To bazw sth 8esh mnhmhs
INX H	;Paw sthn epomenh 8esh mnhmhs
CMP C	;Sygkrinw me ton mexri twra mikrotero ari8mo
JNC STOP	
MOV C,A	;An einai mikroteros, ton bazw sto C
STOP:	;O mE mhdenizei to SOD2 ("phra, anamonh gia neo")
MVI A,00H	
OUT 20H	
JMP START	
END	

Άσκηση 3


Παρέχονται ολοκληρωμένα κυκλώματα: ROMs των 4Kbytes και 16Kbytes, RAMs των 4Kbytes, μE 8085, καταχωρητές και απομονωτές (απλής και διπλής κατεύθυνσης) των 8 bits, αποκωδικοποιητές 3 σε 8 και οι βασικές λογικές πύλες.

Συμβολική ετικέτα	Τύπος	Αρχή διευθύνσεων	Τέλος διευθύνσεων
ROM1	ROM 16K x 8bit	0000 0000 0000 0000 0 HEX	0010 1111 1111 1111 2FFF HEX
RAM1	RAM 4K x 8bit	0011 0000 0000 0000 3000 HEX	0011 1111 1111 1111 3FFF HEX
RAM2	RAM 4K x 8bit	0100 0000 0000 0000 4000 HEX	0100 1111 1111 1111 4FFF HEX
ROM1	ROM 16K x 8bit	0101 0000 0000 0000 5000 HEX	0101 1111 1111 1111 5FFF HEX
ROM2	ROM 4K x 8bit	0110 0000 0000 0000 6000 HEX	0110 1111 1111 1111 6FFF HEX

Η θύρα εισόδου 7000H θα υλοποιηθεί με απεικόνιση μνήμης (και μερική αποκωδικοποίηση).

Η θύρα εξόδου 70H θα υλοποιηθεί με standard IO (και πλήρη αποκωδικοποίηση).

Θυμίζουμε πως $7000\text{H} = 0111\ 0000\ 0000\ 0000\ \text{b}$ και $70\text{H} = 0111\ 0000\ \text{b}$



Άσκηση 4

α)

SWAP_NIBBLE MACRO Q

```
PUSH PSW ;Bazw sth stoiba ton A kai ta flags
MOV A,Q ;Metaferw to periexomeno tou Q ston A
RRC ;Kanw 4 peristrofes,
RRC ;wste na enallaksw ta 4MSB me ta 4LSB
RRC ;(ka8e HEX pshfio einai 4 bit)
RRC
MOV Q,A ;Bazw th nea timh ston Q
POP PSW ;Anaktw thn timh tou A kai tw n flags
```

ENDM

β)

RHLR MACRO Q,R

```
PUSH PSW ;Bazw sth stoiba thn timh tou A kai tw n flags
MOV A,Q ;Bazw to periexomeno tou Q ston A
RAR ;Peristrefw mesw CY: CY->Q7,Q(n)->Q(n-1), Q0->CY
MOV Q,A ;Bazw th nea timh ston Q
MOV A,R ;Omoia ston R
RAR
MOV R,A ;Telika CY->Q7,Q(n)->Q(n-1), Q0->R7, R(n)->R(n-1), R0->CY
JNC CY_0 ;An to CY egine 0, kanoume alma
POP PSW ;Anaktoume thn timh tou A kai tw n flags
STC ;To CY prepei na ginei 1 mesw ths macro, opote to 8etoume
JMP DONE
```

```
CY_0: ;An ftasoume edw, to CY prepei na ginei 0 mesw ths macro
POP PSW ;Anaktoume thn timh tou A kai tw n flags
STC ;8etoume to CY
CMC ;Kai to symplhrwnoume gia na to mhdenisoume
```

DONE:

ENDM